Integralverfahren zur Erstellung der Temperaturfunktionen in einer zylindrischen Säule und ihrer Umgebung

Gerhard Herres, 2025-02-18

> $restart:$

T1 ist die Temperaturverteilung in der zylindrischen Säule, T2 ist die Temperaturverteilung um die Säule herum.
In der Säule wird Wärme homogen freigesetzt. W0=86 W/m³. Durchmesser der Säule ist D=19.4 m, Länge 800 m.

> $T1 := c0 - c2 \cdot r^2;\ T2 := c3 \cdot (r - r0)^2;$

$$T1 := c0 - c2\, r^2$$

$$T2 := c3\, (r - r0)^2 \qquad (1)$$

Grenzbedingungen: 1. T1(Rs,t) = T2(Rs,t) ,   2. diff(T1(r,t),r) = diff(T2(r,t),r) bei r=Rs=9.7 m

> $eq1 := subs(r = Rs,\ T1) = subs(r = Rs,\ T2);$

$$eq1 := c0 - c2\, Rs^2 = c3\, (Rs - r0)^2 \qquad (2)$$

> $eq2 := subs(r = Rs,\ diff(T1, r)) = subs(r = Rs,\ diff(T2, r));$

$$eq2 := -2\, c2\, Rs = 2\, c3\, (Rs - r0) \qquad (3)$$

> $Loes2u3 := solve(\{eq1, eq2\}, \{c2, c3\});$

$$Loes2u3 := \left\{ c2 = \frac{c0}{Rs\, r0},\ c3 = -\frac{c0}{r0\, (Rs - r0)} \right\} \qquad (4)$$

> $C2 := rhs(Loes2u3[1]);\ C3 := rhs(Loes2u3[2]);$

$$C2 := \frac{c0}{Rs\, r0}$$

$$C3 := -\frac{c0}{r0\, (Rs - r0)} \qquad (5)$$

> $T1 := subs(c2 = C2,\ T1);\ T2 := subs(c3 = C3,\ T2);$

$$T1 := c0 - \frac{c0\, r^2}{Rs\, r0}$$

$$T2 := -\frac{c0\, (r - r0)^2}{r0\, (Rs - r0)} \qquad (6)$$

>

> $Tg := piecewise(0 < r < Rs,\ T1,\ r0 > r > Rs,\ T2,\ r > r0,\ 0);$

$$Tg := \begin{cases} c0 - \dfrac{c0\, r^2}{Rs\, r0} & 0 < r \textbf{ and } r < Rs \\ -\dfrac{c0\, (r - r0)^2}{r0\, (Rs - r0)} & r < r0 \textbf{ and } Rs < r \\ 0 & r0 < r \end{cases} \qquad (7)$$

Abfliessende Wärme

> Qab := pi·Rs^2·H·W0·exp(-b·t) =-2·pi·Rs·H·lambda·subs(r=Rs, diff(T1, r));

$$Qab := \pi Rs^2 H W0 e^{-b t} = \frac{4 \pi Rs H \lambda c0}{r0} \quad \mathbf{(8)}$$

Gesamte Wärme

Randbedingung: Integral über beide Bereiche bis zu r=r0 muss gleich sein der insgesamt freigesetzten Wärmemenge.

> eq3 := H·rho·cp·(int(T1·2·pi·r, r=0..Rs) + int(T2·2·pi·r, r=Rs..r0)) = pi·Rs^2·H·W0·int(exp(-b·t), t=0..t0);

$$eq3 := H \rho cp \left(-\frac{1}{2} \frac{c0 Rs^3 \pi}{r0} + c0 \pi Rs^2 - \frac{1}{2} \frac{c0 \pi (r0^4 - Rs^4)}{r0 (Rs - r0)} + \frac{4}{3} \frac{c0 \pi (r0^3 - Rs^3)}{Rs - r0} - \frac{c0 r0 \pi (r0^2 - Rs^2)}{Rs - r0} \right) = -\frac{\pi Rs^2 H W0 (-1 + e^{-b t0})}{b} \quad \mathbf{(9)}$$

> eq3 := normal(lhs(eq3)) = rhs(eq3);

$$eq3 := \frac{1}{6} (Rs^2 + Rs r0 + r0^2) \pi c0 H \rho cp = -\frac{\pi Rs^2 H W0 (-1 + e^{-b t0})}{b} \quad \mathbf{(10)}$$

> C0 := solve(Qab, c0);

$$C0 := \frac{1}{4} \frac{Rs W0 e^{-b t} r0}{\lambda} \quad \mathbf{(11)}$$

> eq3a := subs(c0 = C0, t0 = t, eq3);

$$eq3a := \frac{1}{24} \frac{(Rs^2 + Rs r0 + r0^2) \pi Rs W0 e^{-b t} r0 H \rho cp}{\lambda} = -\frac{\pi Rs^2 H W0 (-1 + e^{-b t})}{b} \quad \mathbf{(12)}$$

> eq3b := normal(lhs(eq3a) - rhs(eq3a));

$$eq3b := \frac{1}{24} \frac{1}{\lambda b} \left(\pi Rs W0 H \left(e^{-b t} r0 \rho cp b Rs^2 + e^{-b t} r0^2 \rho cp b Rs + e^{-b t} r0^3 \rho cp b - 24 Rs \lambda + 24 Rs \lambda e^{-b t}\right)\right) \quad \mathbf{(13)}$$

> tg := solve(eq3b, t);

$$tg := -\frac{\ln\left(\frac{24 Rs \lambda}{r0 \rho cp b Rs^2 + r0^2 \rho cp b Rs + r0^3 \rho cp b + 24 Rs \lambda}\right)}{b} \quad \mathbf{(14)}$$

tg ist die Zeit bis zu der die Wärme bis r=r0 geflossen ist.

> C0g := simplify(subs(t = tg, C0));

$$C0g := \frac{6 Rs^2 W0 r0}{r0 \rho cp b Rs^2 + r0^2 \rho cp b Rs + r0^3 \rho cp b + 24 Rs \lambda} \quad \mathbf{(15)}$$

C0g ist die maximale Temperatur in der Mitte der Säule, wenn die Wärme r0 erreicht hat.

> C0G := (subs(rho = 2200, cp = 1200, b = 5.48849845e-10, W0 = 86.5474, Rs = 9.7, lambda = 5.6, C0g));

$$C0G := \frac{48859.46920 r0}{0.1363329843 r0 + 0.01405494683 r0^2 + 0.001448963591 r0^3 + 1303.68} \quad \mathbf{(16)}$$

> plot(C0G, r0 = 9.7..100);



> $r0gmax := solve(diff(C0G, r0) = 0, r0);$

$$r0gmax := 75.04032112, -39.94516056 + 66.32781844\,I, -39.94516056 - 66.32781844\,I \quad (17)$$

Bei r0=75.04 m ist die Temperatur in der Mitte maximal.

> $C0gmax := subs(r0 = r0gmax[1], C0G);$

$$C0gmax := 1828.349413 \quad (18)$$

> $TG := (subs(\mathrm{rho} = 2200, cp = 1200, b = 5.48849845\mathrm{e}{-10}, W0 = 86.5474, Rs = 9.7, \mathrm{lambda} = 5.6, tg));$

$$TG := -1.821991951\ 10^9 \ln\left(1303.68 \middle/ \left(0.1363329843\, r0 + 0.01405494683\, r0^2 + 0.001448963591\, r0^3 + 1303.68\right)\right) \quad (19)$$

> $plot(TG, r0 = 9.7\,..100);$



> $TGmax := evalf(subs(r0 = r0gmax[1], TG));$

$$TGmax := 7.845750106\ 10^{8} \tag{20}$$

Die maximale Temperatur tritt nach 7.845E8 Sekunden = 24,86 Jahren ein und beträgt 1828 K über der Anfangstemperatur.

> $with(plots);$

$$[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra\_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot] \tag{21}$$

> $Tg0 := subs\left(Rs = 9.7, c0 = \dfrac{48859.46920\ r0}{0.1363329843\ r0 + 0.01405494683\ r0^2 + 0.001448963591\ r0^3 + 1303.68}, Tg\right);$

$$Tg0 := \begin{cases} \dfrac{48859.46920\ r0}{0.1363329843\ r0 + 0.01405494683\ r0^2 + 0.001448963591\ r0^3 + 1303.68} - \dfrac{}{0.1363329843\ r0 + 0.01 \ldots} \\ \qquad - \dfrac{48859.46920\ (r - r0)^2}{(0.1363329843\ r0 + 0.01405494683\ r0^2 + 0.001448963591\ r0^3 + 130 \ldots} \\ \qquad 0 \end{cases}$$

>

> $animate(\ plot\ ,\ [Tg0, r=0..100]\ ,\ r0=9.7..100);$



>

> **for** k **from** 10 **by** 10 **to** 200 **do**
> $p[k] := plot(subs(r0=k,\ Tg0),\ r=0..200)$:
> **end do**:

> $display(p[10], p[20], p[30], p[40], p[50], p[60], p[70]);$



> $display(p[80], p[90], p[100], p[110], p[120], p[130], p[140], p[150], p[160], p[170], p[180], p[190], p[200]);$



> $plot(TG, r0 = 9.7 .. 200);$



Zeit in [s] für r0 [m]
t [s]
4, × 10^9
3, × 10^9
2, × 10^9
1, × 10^9
20
40
60
80
100
120
140
160
180
200
r0 [m]